NAG Library Routine Document

F08UAF (DSBGV)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08UAF (DSBGV) computes all the eigenvalues and, optionally, the eigenvectors of a real generalized symmetric-definite banded eigenproblem, of the form

$$Az = \lambda Bz,$$

where A and B are symmetric and banded, and B is also positive-definite.

2 Specification

```
 SUBROUTINE F08UAF(JOBZ, UPLO, N, KA, KB, AB, LDAB, BB, LDBB, W, Z, LDZ,
1                  WORK, INFO)
 INTEGER           N, KA, KB, LDAB, LDBB, LDZ, INFO
 double precision  AB(LDAB,*), BB(LDBB,*), W(N), Z(LDZ,*), WORK(3*N)
 CHARACTER*1       JOBZ, UPLO
```

The routine may be called by its LAPACK name ***dsbgv***.

3 Description

The generalized symmetric-definite band problem

$$Az = \lambda Bz$$

is first reduced to a standard band symmetric problem

$$Cx = \lambda x,$$

where C is a symmetric band matrix, using Wilkinson's modification to Crawford's algorithm (see Crawford (1973) and Wilkinson (1977)). The symmetric eigenvalue problem is then solved for the eigenvalues and the eigenvectors, if required, which are then backtransformed to the eigenvectors of the original problem.

The eigenvectors are normalized so that the matrix of eigenvectors, Z, satisfies

$$Z^{\mathrm{T}}AZ = \Lambda \qquad \text{and} \qquad Z^{\mathrm{T}}BZ = I,$$

where Λ is the diagonal matrix whose diagonal elements are the eigenvalues.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphiahttp://www.netlib.org/lapack/lug

Crawford C R (1973) Reduction of a band-symmetric generalized eigenvalue problem *Comm. ACM* **16** 41–44

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Wilkinson J H (1977) Some recent advances in numerical linear algebra *The State of the Art in Numerical Analysis* (ed D A H Jacobs) Academic Press

5 Parameters

1: JOBZ – CHARACTER*1 *Input*

On entry: if JOBZ = 'N', compute eigenvalues only.

If JOBZ = 'V', compute eigenvalues and eigenvectors.

Constraint: JOBZ = 'N' or 'V'.

2: UPLO – CHARACTER*1 *Input*

On entry: if UPLO = 'U', the upper triangles of A and B are stored.

If UPLO = 'L', the lower triangles of A and B are stored.

Constraint: UPLO = 'U' or 'L'.

3: N – INTEGER *Input*

On entry: n, the order of the matrices A and B.

Constraint: $\mathrm{N} \geq 0$.

4: KA – INTEGER *Input*

On entry: if UPLO = 'U', the number of superdiagonals, k_a, of the matrix A.

If UPLO = 'L', the number of subdiagonals, k_a, of the matrix A.

Constraint: $\mathrm{KA} \geq 0$.

5: KB – INTEGER *Input*

On entry: if UPLO = 'U', the number of superdiagonals, k_b, of the matrix B.

If UPLO = 'L', the number of subdiagonals, k_b, of the matrix B.

Constraint: $\mathrm{KA} \geq \mathrm{KB} \geq 0$.

6: AB(LDAB,$*$) – ***double precision*** array *Input/Output*

Note: the second dimension of the array AB must be at least $\max(1,\mathrm{N})$.

On entry: the upper or lower triangle of the n by n symmetric band matrix A.

The matrix is stored in rows 1 to $k_a + 1$, more precisely,

if UPLO = 'U', the elements of the upper triangle of A within the band must be stored with element A_{ij} in $\mathrm{AB}(k_a + 1 + i - j, j)$ for $\max(1, j - k_a) \leq i \leq j$;

if UPLO = 'L', the elements of the lower triangle of A within the band must be stored with element A_{ij} in $\mathrm{AB}(1 + i - j, j)$ for $j \leq i \leq \min(n, j + k_a)$.

On exit: the contents of AB are destroyed.

7: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F08UAF (DSBGV) is called.

Constraint: $\mathrm{LDAB} \geq \mathrm{KA} + 1$.

8: BB(LDBB,$*$) – ***double precision*** array *Input/Output*

Note: the second dimension of the array BB must be at least $\max(1,\mathrm{N})$.

On entry: the upper or lower triangle of the n by n symmetric band matrix B.

The matrix is stored in rows 1 to $k_b + 1$, more precisely,

if UPLO = 'U', the elements of the upper triangle of B within the band must be stored with element B_{ij} in BB$(k_b + 1 + i - j, j)$ for $\max(1, j - k_b) \leq i \leq j$;

if UPLO = 'L', the elements of the lower triangle of B within the band must be stored with element B_{ij} in BB$(1 + i - j, j)$ for $j \leq i \leq \min(n, j + k_b)$.

On exit: the factor S from the split Cholesky factorization $B = S^{\mathrm{T}}S$, as returned by F08UFF (DPBSTF).

9: LDBB – INTEGER *Input*

On entry: the first dimension of the array BB as declared in the (sub)program from which F08UAF (DSBGV) is called.

Constraint: LDBB $\geq$ KB + 1.

10: W(N) – ***double precision*** array *Output*

On exit: if INFO = 0, the eigenvalues in ascending order.

11: Z(LDZ,$*$) – ***double precision*** array *Output*

Note: the second dimension of the array Z must be at least $\max(1, \mathrm{N})$.

On exit: if JOBZ = 'V', then if INFO = 0, Z contains the matrix Z of eigenvectors, with the ith column of Z holding the eigenvector associated with W(i). The eigenvectors are normalized so that $Z^{\mathrm{T}}BZ = I$.

If JOBZ = 'N', Z is not referenced.

12: LDZ – INTEGER *Input*

On entry: the first dimension of the array Z as declared in the (sub)program from which F08UAF (DSBGV) is called.

Constraints:

if JOBZ = 'V', LDZ $\geq \max(1, \mathrm{N})$;
otherwise LDZ ≥ 1.

13: WORK($3 \times$ N) – ***double precision*** array *Workspace*

14: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$ and $i \leq$ N, the algorithm failed to converge: i off-diagonal elements of an intermediate tridiagonal form did not converge to zero.

If INFO $= i$ and $i >$ N, if INFO = N $+ i$, for $1 \leq i \leq$ N, then F08UFF (DPBSTF) returned INFO $= i$: B is not positive-definite. The factorization of B could not be completed and no eigenvalues or eigenvectors were computed.

7 Accuracy

If B is ill-conditioned with respect to inversion, then the error bounds for the computed eigenvalues and vectors may be large, although when the diagonal elements of B differ widely in magnitude the eigenvalues and eigenvectors may be less sensitive than the condition of B would suggest. See Section 4.10 of Anderson *et al.* (1999) for details of the error bounds.

8 Further Comments

The total number of floating-point operations is proportional to n^3 if JOBZ = 'V' and, assuming that $n \gg k_a$, is approximately proportional to $n^2 k_a$ otherwise.

The complex analogue of this routine is F08UNF (ZHBGV).

9 Example

This example finds all the eigenvalues of the generalized band symmetric eigenproblem $Az = \lambda Bz$, where

$$A = \begin{pmatrix} 0.24 & 0.39 & 0.42 & 0 \\ 0.39 & -0.11 & 0.79 & 0.63 \\ 0.42 & 0.79 & -0.25 & 0.48 \\ 0 & 0.63 & 0.48 & -0.03 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 2.07 & 0.95 & 0 & 0 \\ 0.95 & 1.69 & -0.29 & 0 \\ 0 & -0.29 & 0.65 & -0.33 \\ 0 & 0 & -0.33 & 1.17 \end{pmatrix}.$$

9.1 Program Text

```
*     F08UAF Example Program Text
*     Mark 21 Release.  NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, KAMAX, KBMAX
      PARAMETER        (NMAX=20,KAMAX=5,KBMAX=5)
      INTEGER          LDAB, LDBB
      PARAMETER        (LDAB=KAMAX+1,LDBB=KBMAX+1)
      CHARACTER        UPLO
      PARAMETER        (UPLO='U')
*     .. Local Scalars ..
      INTEGER          I, INFO, J, KA, KB, N
*     .. Local Arrays ..
      DOUBLE PRECISION AB(LDAB,NMAX), BB(LDBB,NMAX), DUMMY(1,1),
     +                 W(NMAX), WORK(3*NMAX)
*     .. External Subroutines ..
      EXTERNAL         DSBGV
*     .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08UAF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KA, KB
      IF (N.LE.NMAX .AND. KA.LE.KAMAX .AND. KB.LE.KBMAX) THEN
*
*        Read the upper or lower triangular parts of the matrices A and
*        B from data file
*
         IF (UPLO.EQ.'U') THEN
            READ (NIN,*) ((AB(KA+1+I-J,J),J=I,MIN(N,I+KA)),I=1,N)
            READ (NIN,*) ((BB(KB+1+I-J,J),J=I,MIN(N,I+KB)),I=1,N)
         ELSE IF (UPLO.EQ.'L') THEN
            READ (NIN,*) ((AB(1+I-J,J),J=MAX(1,I-KA),I),I=1,N)
            READ (NIN,*) ((BB(1+I-J,J),J=MAX(1,I-KB),I),I=1,N)
         END IF
*
*        Solve the generalized symmetric band eigenvalue problem
*        A*x = lambda*B*x
*
```

```
         CALL DSBGV('No vectors',UPLO,N,KA,KB,AB,LDAB,BB,LDBB,W,DUMMY,1,
     +              WORK,INFO)
*
         IF (INFO.EQ.0) THEN
*
*           Print solution
*
            WRITE (NOUT,*) 'Eigenvalues'
            WRITE (NOUT,99999) (W(J),J=1,N)
         ELSE IF (INFO.GT.N .AND. INFO.LE.2*N) THEN
            I = INFO - N
            WRITE (NOUT,99998) 'The leading minor of order ', I,
     +        ' of B is not positive definite'
         ELSE
            WRITE (NOUT,99997) 'Failure in DSBGV. INFO =', INFO
         END IF
      ELSE
         WRITE (NOUT,*) 'NMAX too small'
      END IF
*
99999 FORMAT (3X,(6F11.4))
99998 FORMAT (1X,A,I4,A)
99997 FORMAT (1X,A,I4)
      END
```

9.2 Program Data

```
F08UAF Example Program Data

  4      2      1            :Values of N, KA and KB

  0.24   0.39   0.42
        -0.11   0.79   0.63
               -0.25   0.48
                      -0.03  :End of matrix A
  2.07   0.95
         1.69  -0.29
                0.65  -0.33
                       1.17  :End of matrix B
```

9.3 Program Results

```
 F08UAF Example Program Results

 Eigenvalues
       -0.8305    -0.6401     0.0992     1.8525
```